Microchip PICBASIC PRO Firmware - RS-485 network demo

' FileName:        rs485.bas

' Processor:       PIC18

' Compiler:        PICBASIC PRO

' Software License Agreement

'

' Licensor grants any person obtaining a copy of this software ("You")

' a worldwide, royalty-free, non-exclusive license, for the duration of

' the copyright, free of charge, to store and execute the Software in a

' computer system and to incorporate the Software or any portion of it

' in computer programs You write.

' THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR

' IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,

' FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE

' AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER

' LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,

' OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN

' THE SOFTWARE.

' Author               Date        Comment

' Jan Axelson          10/20/07    Original

'************************************************************************

' The PIC is a secondary node in an RS-485 network. The USART interfaces

' to an RS-485 transceiver. A network can have multiple secondary nodes.

' The PIC responds to defined commands. Each command has this format:

' byte 1: ":"

' byte 2: the PIC's network address

' byte 3: command code

' bytes 4..n: additional data (command-dependent)

' final byte: LF code (0ah)

' Each response has this format:

' byte 1: ":"

' byte 2: the PIC's network address

' byte 3..n: additional data (command dependent)

' final byte: LF code (0ah)

' This example defines two commands, read_byte and write_byte.

' To read a byte, send this command to the PIC:

' :h2b

' followed by a LF code (hit Enter)

' The PIC will respond with:

' :hxx

' followed by a LF, where xx is the hex value read.

' To write a byte, send this command to the PIC:

' :h1bxx

' followed by a LF, where xx is the hex value to write.

' (For example, :h1bff or :h1b00)

' The PIC will respond with:

' :h

' followed by a LF.

' A PC can communicate with the PIC using a custom application, or you can use

' a terminal-emulator such as HyperTerminal if the PC's driver-enable line is

' controlled entirely in hardware.

' On the PIC, firmware can control the RS-485 driver-enable line if needed.

' This application and host applications to communicate with the device are available

' from www.Lvr.com.

'************************************************************************

' Serial port configuration

DEFINE HSER_RCSTA 90h

DEFINE HSER_TXSTA 24h

DEFINE HSER_SPBRG 19h

DEFINE HSER_CLROERR

COMMAND_START           con     ":"

MAX_COMMAND_LENGTH      con     5

MY_ADDRESS              con     "h"

' Use any spare output port bit:

driver_enable                   var     PORTB.3  'UNNEEDED IF USING HARDWARE-ONLY DRIVER ENABLE

command_index                   var     byte

command_response                var     byte[6]

converted_byte                  var     byte

data_ready_to_send              var     bit

delay_before_responding var bit

firmware_driver_enable  var     bit

index                                   var byte

lower_nibble                    var     byte

network_state                   var byte

received_command                var     byte[6]

response_index                  var     byte

serial_in                               var     byte

serial_out                              var byte

success                                 var     bit

upper_nibble                    var     byte

value_to_convert                var     byte

command_index = 0

command_response [0] = COMMAND_START

command_response [1] = MY_ADDRESS

network_state = "r"

response_index = 0

' Output port bit used in testing (write_byte command):

TRISB 0# = 0

low PORTB.0

' Uncomment one of these lines:

'delay_before_responding = 0  ' No delay before responding.

delay_before_responding = 1  ' Delay before responding

' Uncomment one of these lines:

firmware_driver_enable = 0  ' Circuits have automatic driver-enable control.

'firmware_driver_enable = 1  ' Firmware must control the driver enable.

If (firmware_driver_enable = 1) Then

        ' Configure the driver-enable line and disable the driver.

        TRISB 0.3 = 0

        driver_enable = 0

End If

Loop:

        ' The main program loop

        GoSub serial_communications

        ' Add other tasks here.

goto loop

'************************************************************************

' Routine:              ascii_hex_to_byte

' PreCondition:         None

' Input:                upper_nibble - ASCII hex code for the upper 4 bits of a byte

'                                       lower_nibble - ASCII hex code for the lower 4 bits of a byte

' Output:               converted_byte - the value represented by upper_nibble

'                                       and lower_nibble

'                                       success - indicates conversion success (1) or failure (0)

'

' Side Effects:         None

' Overview:             Converts 2 ASCII hex characters

'                                       to the byte value they represent.

' Note:                 Assumes hex characters a-f are lower case.

'************************************************************************

ascii_hex_to_byte:

        success = 1

        ' Convert each character code to the value it represents.

        Select Case upper_nibble

                Case "0", "1", "2", "3", "4", "5", "6", "7", "8", "9"

                        upper_nibble = upper_nibble - 48

                Case "a", "b", "c", "d", "e", "f"

                        upper_nibble = upper_nibble - 87

                Case "A", "B", "C", "D", "E", "F"

                        upper_nibble = upper_nibble - 55

                Case Else

                        ' The text character isn't 0-9, a-f, or A-F.

                        success = 0

        End Select

        Select Case lower_nibble

                Case "0", "1", "2", "3", "4", "5", "6", "7", "8", "9"

                        lower_nibble = lower_nibble - 48

                Case "a", "b", "c", "d", "e", "f"

                        lower_nibble = lower_nibble - 87

                Case "A", "B", "C", "D", "E", "F"

                        lower_nibble = lower_nibble - 55

                Case Else

                        ' The text character isn't 0-9, a-f, or A-F.

                        success = 0

        End Select

        If (success = 1) Then

                ' Combine the nibbles in a byte.

                converted_byte = (upper_nibble << 4) + lower_nibble

        End If

Return

'************************************************************************

' Routine:              byte_to_ascii_hex

' PreCondition:         None

' Input:                value_to_convert - the byte value to convert

' Output:               upper_nibble - character code representing the upper 4 bits

'                                       lower_nibble - character code representing the lower 4 bits

' Side Effects:         None

' Overview:             Convert a byte to 2 ASCII hex characters

' Note:

'************************************************************************

byte_to_ascii_hex:

        upper_nibble = (value_to_convert & $f0) >> 4

        If ((upper_nibble >= 0) And (upper_nibble <= 9)) Then

                        upper_nibble = upper_nibble + 48

        Else

                ' The value is between 10 (a) and 15 (f).

                upper_nibble = upper_nibble + 87

        End If

        lower_nibble = (value_to_convert & $0f)

        If ((lower_nibble >= 0) And (lower_nibble <= 9)) Then

                        lower_nibble = lower_nibble + 48

        Else

                ' The value is between 10 (a) and 15 (f).

                lower_nibble = lower_nibble + 87

        End If

Return

'************************************************************************

' Routine:              check_response_delay

' PreCondition:         delay_before_responding indicates whether the PIC

'                                       must delay before responding to a received command.

' Input:                None

' Output:               If the delay has timed out, sets network_state = "t"

' Side Effects:         None

' Overview:             If a delay before responding is required, check the

'                                       timer. If the delay has timed out, enable responding.

' Note:

'************************************************************************

check_response_delay:

        If (delay_before_responding = 1) Then

                if (INTCON.2 = 1) then

                        ' The delay time has elapsed. Stop the timer and send the response.

                        T0CON = 0

                        INTCON 0.2 = 0

                        network_state = "t"

                End If

        End If

Return

'************************************************************************

' Routine:              command_read_byte

' PreCondition:         A read_byte command has been received.

' Input:                received_command[] - the command

' Output:               command_response[] - the response

' Side Effects:         None

' Overview:             Processes a read_byte command and prepares to respond.

' Note:

'************************************************************************

command_read_byte:

        ' A read_byte command has been received.

        select case received_command[3]

                ' Add more cases (locations to read) as needed.

                Case "b"

                        ' Read Port B.

                        value_to_convert = PORTB

                        ' Convert the value read to ASCII Hex

                        GoSub byte_to_ascii_hex

                        ' Prepare to send the ASCII Hex characters in a response.

                        command_response [2] = upper_nibble

                        command_response [3] = lower_nibble

                        command_response[4] = $0a

                        GoSub prepare_to_respond

                Case Else

        End Select

        Return

'************************************************************************

' Routine:              command_write_byte

' PreCondition:         A write_byte command has been received.

' Input:                received_command[] - the command

' Output:               command_response[] - the response

' Side Effects:         None

' Overview:             Processes a write_byte command and prepares to respond.

' Note:                 None

'************************************************************************

command_write_byte:

        select case received_command[3]

                ' Add more cases as needed.

                Case "b"

                        ' Get the data to write.

                        ' Convert the received ASCII Hex bytes to a byte value.

                        upper_nibble = received_command[4]

                        lower_nibble = received_command[5]

                        GoSub ascii_hex_to_byte

                        ' Set bit 0 of PortB to match bit 0 in the received byte.

                        If ((converted_byte & 1) = 1) Then

                                high PORTB.0

                        Else

                                low PORTB.0

                        End If

                        command_response[2] = $0a

                        GoSub prepare_to_respond

                Case Else

        End Select

Return

'************************************************************************

' Routine:              initialize_serial_buffers

' PreCondition:         None

' Input:                None

' Output:               None

' Side Effects:         None

' Overview:             initalize variables relating to serial data

' Note:

'************************************************************************

initialize_serial_buffers:

        command_index = 0

        response_index = 0

        received_command [0] = 0

Return

'************************************************************************

' Routine:              prepare_to_respond

' PreCondition:         The PIC has a response to send.

'                                       delay_before_responding indicates whether to delay (1)

'                                       or not (0) before sending the response.

' Input:                None

' Output:               network_state is set to "d" or "t"

' Side Effects:         None

' Overview:             Prepares to respond to a command.

'                                       If a delay is needed, sets network_state = "d".

'                       Otherwise sets network_state = "t".

' Note:

'************************************************************************

prepare_to_respond:

        response_index = 0;

        If (delay_before_responding = 1) Then

                GoSub start_response_delay_timer

                network_state = "d"

        Else

                network_state = "t"

        End If

Return

'************************************************************************

' Routine:              receive_serial_data

' PreCondition:         An open serial port. See DEFINE HSER_* statements

' Input:                None

' Output:               received_command[] stores received command bytes

'

' Side Effects:         None

' Overview:             Processes received bytes.

' Note:                 None

'************************************************************************

receive_serial_data:

        ' Process received bytes.

        if (PIR1.5 = 1) then

                ' A byte is available to read.

                if (RCSTA.2 = 1) then

                        ' Framing error. Read RCREG to clear the error

                        ' but don't use the data.

                        hserin [serial_in]

                Else

                        ' No framing error occurred

                        hserin [serial_in]

                        Select Case serial_in

                                case $0a

                                        ' A LF character was received, indicating the end of a command.

                                        GoSub respond_to_command

                                case $0d

                                        ' Ignore a received CR character.

                                Case COMMAND_START

                                        ' A new command has begun.

                                        ' Initialize the array that holds received bytes.

                                        received_command [0] = COMMAND_START

                                        command_index = 1

                                Case Else

                                        ' A character was received and it's not a LF or CR.

                                        ' The primary node might send extraneous data.

                                        ' If at the end of the array, ignore additional received data.

                                        If (command_index <= MAX_COMMAND_LENGTH) Then

                                                ' Convert characters A-Z to lower case.

                                                If ((serial_in >= "A") And (serial_in <= "Z")) Then

                                                        serial_in = serial_in + 32

                                                End If

                                                ' Save the character and increment the position in the array

                                                ' that stores received text.

                                                received_command [command_index] = serial_in

                                                command_index = command_index + 1

                                        End If

                        End Select

                End If

        End If

Return

'************************************************************************

' Routine:              respond_to_command

' PreCondition:         None

' Input:                None

' Output:               None

'

' Side Effects:         None

' Overview:             Check for a valid, supported command and

'                                       if detected, call a routine to handle the command.

' Note:                 None

'************************************************************************

respond_to_command:

        ' Every command begins with a COMMAND_START code, the node's address,

        ' and a command code.

        if (received_command[0] = COMMAND_START)  then

                if (received_command[1] = MY_ADDRESS) then

                        select case received_command[2]

                                Case "1"

                                        GoSub command_write_byte

                                Case "2"

                                        GoSub command_read_byte

                                Case Else

                                        GoSub initialize_serial_buffers

                        End Select

                Else

                        GoSub initialize_serial_buffers

                End If

        Else

                GoSub initialize_serial_buffers

        End If

Return

'************************************************************************

' Routine:              serial_communications

' PreCondition:         None

' Input:                None

' Output:               None

' Side Effects:         None

' Overview:             Implements a state machine for serial communications tasks.

' Note:

'************************************************************************

serial_communications:

        Select Case network_state

                Case "r"

                        GoSub receive_serial_data

                Case "d"

                        GoSub check_response_delay

                Case "t"

                        GoSub transmit_serial_data

                Case Else

        End Select

Return

'************************************************************************

' Routine:              start_response_delay_timer

' PreCondition:         None

' Input:                None

' Output:               None

' Side Effects:         Uses TIMER0

' Overview:             Starts a timer to implement a delay before responding.

' Note:

'************************************************************************

start_response_delay_timer:

        ' This example sets a delay of around 0.5 sec. assuming FOSC = 4 Mhz.

        ' Timer enabled, 16-bit, internal clock, prescaler = 256.

        T0CON = $87

        ' Load the timer with F800h.

        ' Both bytes load on a write to TMR0L, so set the value of TMR0H first.

        TMR0H = $F8

        TMR0L = $00

Return

'************************************************************************

' Routine:              transmit_serial_data

' PreCondition:         An open serial port. See DEFINE HSER_* statements

' Input:                None

' Output:               None

'

' Side Effects:         None

' Overview:             If a byte is waiting to transmit, send it.

' Note:                 None

'************************************************************************

transmit_serial_data:

                If firmware_driver_enable Then

                        driver_enable = 1

                End If

                ' Wait for the transmit shift register to empty.

                while (TXSTA.1 = 0)

                Wend

                hserout [command_response[response_index]]

                if (command_response[response_index] = $0a) then

                        ' The entire response has been sent.

                        If firmware_driver_enable Then

                                while (TXSTA.1 = 0)

                                Wend

                                driver_enable = 0

                        End If

                        GoSub initialize_serial_buffers

                        network_state = "r"

                Else

                        ' Prepare to send another byte.

                        response_index = response_index + 1

                End If

Return

